


2012

# UNIDAD V: ESTRUCTURAS REPETITIVAS



## UNIDAD V: ESTRUCTURAS REPETITIVAS O CICLICAS

### 5.1 INTRODUCCIÓN:

Hasta el momento se ha encontrado que cada una de las instrucciones que conforman el algoritmo se ejecutan una, y sólo una vez, en el mismo orden en que aparecían. Los algoritmos de este tipo son realmente simples, ya que no incluyen una estructura que permita que un grupo de instrucciones se ejecute varias veces, como resultado de una determinada condición.

La mayoría de los problemas dentro de la programación exigen que un grupo de instrucciones que hacen un cálculo determinado no se hagan para un ente específico, sino que sea aplicado a muchos para realizar el mismo cálculo. Por ejemplo: el cálculo del promedio de créditos de los estudiantes de una universidad. En este caso es necesario el establecimiento de un modelo que obtenga el promedio para uno y que a la vez sirva para todos los estudiantes, es decir, que el grupo de pasos que conforman el modelo de solución de un determinado proceso dentro del algoritmo, se repita tantas veces como sea necesario.

### 5.2 ESTRUCTURA REPETITIVA PARA:

Es una estructura lógica que permite formar un ciclo repetitivo, el cual es controlado por un contador que tiene que definirse con un valor inicial, un valor final y un incremento. Esto significa que debe conocerse de antemano el número de veces que se debe repetir el ciclo.

Este tipo de ciclo se **repite N veces**. Es utilizada cuando se tiene un esquema cuantitativo.

Por ejemplo cuando se necesita calcular la nómina total de la empresa, se suman los sueldos de los N empleados de la misma. Cuando se necesita obtener el promedio de calificaciones de un curso, se suman las N calificaciones de los alumnos y dividir esa suma entre N, es decir, se sabe de antemano cuántas veces se debe repetir una determinada operación, acción ó tarea.

El número de veces se obtiene del planteamiento del problema o de una lectura que indica que el número de iteraciones se debe realizar para N ocurrencias.

Representación:

```
Para variable_numerica <-valor_inicial Hasta valor_final Con Paso paso Hacer
|
|   secuencia_de_acciones
|
FinPara
```

Donde:

- **variable\_numerica:** es el identificador de la variable que se utilizará para iterar.
- **valor\_inicial:** es el valor de variable\_numérica desde el cual se comenzará a iterar.
- **valor\_final:** es el valor final de variable\_numérica hasta el cual se continuará iterando.

- **paso:** es el incremento o decremento que se debe realizar a variable\_numérica en cada iteración.
- **secuencia\_de\_acciones:** es la lista de instrucciones que desea repetir en cada iteración.

**Ejemplo #1:** Realice un algoritmo que visualice los número del 1 al 10.

**Datos de Entrada:** No hay.

**Datos de salida:** Impresión de los números desde el 1 al 10.

**Definición de variables:**

**i :** contador utilizado para generar los números del 1 al 10.

**Solución:**

**Proceso numeros**

Para i<-1 hasta 10 Con Paso 1 Hacer  
Escribir i;  
FinPara

**FinProceso**

```

** Ejecucion Iniciada. **
1
2
3
4
5
6
7
8
9
10
** Ejecucion Finalizada. **
    
```

**Ejemplo #2:** Realice un pseudocódigo que imprima los números del 1 al 10 con incrementos de 0.5.

**Proceso incremento**

Definir var como real;  
para var<-1 hasta 10 con paso 0.5 Hacer  
Escribir var;  
FinPara

**Finproceso**

```

1
1.5
2
2.5
3
3.5
4
4.5
5
5.5
6
6.5
7
7.5
8
8.5
9
9.5
10
    
```

**Ejemplo #3:** Realice un pseudocódigo para calcular la suma de las calificaciones de un grupo de N estudiantes del curso de Lógica de Programación.

**Proceso sumanotas**

Definir nota,suma\_not,nest Como Entero;  
suma\_not<-0;  
Escribir '¿Cuantos estudiantes?';  
Leer nest;  
para e<-1 hasta nest Hacer  
Escribir 'Nota ',e','';  
Leer nota;  
suma\_not<-suma\_not+nota;  
FinPara  
Escribir 'La suma de las notas es: ',suma\_not;


**FinProceso**

**Ejemplo #4:** Realice un pseudocódigo para calcular el promedio de las calificaciones de un grupo de N estudiantes del curso de Lógica de Programación.

#### Proceso promedionotas

```
Definir prom como real;
Definir e,nota,suma_noto,nest Como Entero;
suma_noto<-0;
Escribir '¿Cuántos estudiantes?';
Leer nest;
Para e<-1 hasta nest con paso 1 Hacer
    Escribir 'Nota ',e,'.';
    Leer nota;
    suma_noto<-suma_noto+nota;
FinPara
prom <- suma_noto/nest;
Escribir 'El promedio de notas es: ',prom;
```

**FinProceso**




```
¿Cuántos estudiantes?
> 5
Nota 1:
> 56
Nota 2:
> 78
Nota 3:
> 98
Nota 4:
> 23
Nota 5:
> 15
El promedio de notas es: 54
```

**Ejemplo #5:** Realice un pseudocódigo para calcular el factorial de un número entero positivo. Ejemplo:  $3! = 1 \cdot 2 \cdot 3 = 6$

#### Proceso factorial

```
Definir facto como entero;
facto<-1;
Escribir 'Ingrese el número: ';
Leer N;
Si N>=0 entonces
    Si N=0 O N=1 entonces
        facto<-facto;
    Sino
        Para contador<-1 Hasta N con paso 1 Hacer
            facto<-facto*contador;
        FinPara
    finSi
    Escribir N,'!= ',facto;
Sino
    Escribir 'No se puede calcular el factorial de números negativos';
FinSi
```

**FinProceso**



```
C:\Archivos de programa\PSelnt\pselnt.exe
*** Ejecucion Iniciada.
Ingrese el numero:
> 5
5!= 120
```

**Ejemplo #6:** Realice un pseudocódigo para calcular el porcentaje de aprobados y reprobados de los N estudiantes del curso de Lógica de Programación. Se sabe que la nota mínima para aprobar es 70 pts.

#### Proceso porcentapro\_rep

```
Definir pap,prep como real;
Definir Nest,e,ap,re,nota como entero;
Escribir '¿Cuántos estudiantes?';
Leer Nest;
Para e<-1 hasta Nest Hacer
    Escribir 'Introduzca la nota ',e;
    leer nota;
    si nota>=70 entonces
```

```

        ap<-ap+1;
    Sino
        re<-re+1;
    FinSi
FinPara
pap<-(ap*100)/Nest;
prep<-(re*100)/Nest;
Escribir 'Hay ',ap ' estudiantes aprobados y representan el ',pap,'%';
Escribir 'Hay ',re ' estudiantes reprobados y representan el ',prep,'%';

```

**FinProceso**

**Ejemplo #7:** Realice un pseudocódigo que reciba como dato de entrada un entero N y calcule e imprima el resultado de la siguiente serie:  $1^1 + 2^2 + 3^3 + 4^4 \dots + N^N$ .

**Proceso serie2**

```

Definir N,sumaserie,s como entero;
sumaserie<-0;
Escribir 'Valor de N: ';
Leer N;
Para s<-1 Hasta N Hacer
    sumaserie<-sumaserie+(s^s);
Finpara
Escribir 'El resultado de la serie es: ',sumaserie;

```

**FinProceso**

**Ejemplo #8:** El gerente de Farmacia “La Baratera” desea un pseudocódigo que calcule el aumento de sueldo para su grupo de N empleados, teniendo en cuenta los siguientes criterios:

- Si el sueldo es inferior a C\$5000 el aumento será del 10%.
- Si el sueldo es superior a C\$5000 el aumento será del 8%.

Imprima el nuevo sueldo del trabajador y el total de nómina de la empresa, considerando el aumento de cada trabajador.

**Proceso farmaciabaratera**

```

Definir Nemp,e como entero;
definir totalnomina,sueldo,i,aumento como real;
Escribir 'Cuántos empleados?';
Leer Nemp;
Para e<-1 hasta Nemp con paso 1 Hacer
    Escribir 'Sueldo del empleado No.',e;
    Leer sueldo;

    si sueldo <5000 entonces
        aumento<- sueldo +(sueldo*0.10);
    Sino
        aumento<- sueldo +(sueldo*0.08);
    FinSi

    Escribir 'El empleado ',e,' ganara ',aumento;
    totalnomina<-totalnomina+aumento;
FinPara

Escribir 'El total de la nomina es C$:',totalnomina;

```

**FinProceso**

**Ejemplo #9:** Elabore un pseudocódigo que calcule e imprima el producto y la suma de los números pares e impares comprendidos entre 1 y 50.

**proceso paresimpares**

```

Definir sumap,prodpar,sumaimp,prodimp,i Como Entero;
prodpar<-1;
prodimpar <- 1;
Para i<-1 Hasta 50 Hacer
    si i MOD 2 = 0 entonces
        sumap<-sumap+i;
        prodpar <- prodpar*i;
    Sino
        sumaimp<-sumaimp +i;
        prodimpar <- prodimpar*i;
    FinSi
FinPara

Escribir 'La suma de los numeros pares entre 1 y 50 es: ',sumap;
Escribir 'El producto de los numeros pares entre 1 y 50 es: ',prodpar;
Escribir 'La suma de los numeros impares entre 1 y 50 es: ',sumaimp;
Escribir 'El producto de los numeros impares entre 1 y 50 es: ',prodimpar;

```

**FinProceso**

**Ejemplo #10:** Elabore un pseudocódigo que lea N números enteros e imprima cuántos son positivos, negativos o nulos.

**Proceso pos\_neg\_nulo**

```

Definir ndatos,NUM,n como entero;
Escribir 'Cuantos numeros?';
Leer ndatos;
Para n<-1 Hasta ndatos con paso 1 Hacer
    Escribir 'Introduzca un número: ';
    Leer NUM;
    si NUM > 0 entonces
        pos<-pos+1;
    sino si NUM < 0 Entonces
        neg<-neg+1;
    Sino
        nul<-nul+1;
    FinSi
FinSi
FinPara
Escribir 'De los ',ndatos,' numeros leidos ',pos,' son POSITIVOS';
Escribir 'De los ',ndatos,' numeros leidos ',neg,' son NEGATIVOS';
Escribir 'De los ',ndatos,' numeros leidos ',nul,' son NULOS';

```

**FinProceso**

**EJERCICIOS PROPUESTOS:**

1) En una tienda de descuento las personas que van a pagar el importe de su compra llegan a la caja y sacan una bolita de color, que les dirá que descuento tendrán sobre el total de su compra. Determinar la cantidad que pagara cada cliente desde que la tienda abre hasta que cierra. Se sabe que si el color de la bolita es roja el cliente obtendrá un 40% de descuento; si es amarilla un 25% y si es blanca no obtendrá descuento.

2) El teatro José de la Cruz Mena otorga descuentos según la edad del cliente. Determinar la cantidad de dinero que el teatro deja de percibir por cada una de las categorías. Tomar en cuenta que los niños menores de 5 años no pueden entrar al teatro y que existe un precio único en los asientos. Los descuentos se hacen tomando en cuenta el siguiente cuadro:

|             | Edad           | Descuento |
|-------------|----------------|-----------|
| Categoría 1 | 5 - 14         | 35 %      |
| Categoría 2 | 15 - 19        | 25 %      |
| Categoría 3 | 20 - 45        | 10 %      |
| Categoría 4 | 46 - 65        | 25 %      |
| Categoría 5 | 66 en adelante | 35 %      |

3) Cierta universidad tiene N estudiantes. Elabore un algoritmo que encuentre el promedio de edad de los estudiantes mayores de 21 años y el promedio de edad del resto de estudiantes. Por cada estudiante se tiene un registro que contiene su código y edad.

4) Elabore un algoritmo que obtenga el cuadrado de los primeros N números enteros.

5) Determinar la cantidad semanal de dinero que recibirá cada uno de los n obreros de una empresa. Se sabe que cuando las horas que trabajo un obrero exceden de 40, el resto se convierte en horas extras que se pagan al doble de una hora normal, cuando no exceden de 8; cuando las horas extras exceden de 8 se pagan las primeras 8 al doble de lo que se paga por una hora normal y el resto al triple.

6) Leer los 250,000 votos otorgados a los 3 candidatos a alcalde de la ciudad de León e imprimir el número del candidato ganador y su cantidad de votos.

7) Elabore un algoritmo que lea 100 números y determine la media de los números positivos y negativos ingresados.

8) Diseñe un pseudocódigo que reciba como dato un entero X y calcule e imprima el resultado de la siguiente serie:  $1^2 - 2^2 + 3^2 - 4^2 \pm N^2$